


Application Note AN3201-02: Pitch Shifting
by Frank Thomson

Introduction

This application note describes a basic pitch shifting routine that can be implemented in the DRE. The DRE is capable of performing pitch shifting by either reading faster (pitch up) or slower (pitch down) through the stored samples. This is accomplished by running the samples through a circular buffer and using two pointers to read from the circular buffer. The two pointers are 180° apart in the circular buffer so that as one pointer is crossing from the end to the start of the buffer, the other pointer is mid-way through the buffer. This is done with two sawtooth waveforms, the discontinuity being where the end-to-start crossing occurs for the pointer.

A sample is calculated by first linearly interpolating between the samples at one pointer, then at the other pointer, then crossfading between the 2 pointers such that the crossfade coefficient is 0 for the pointer as it crosses from the end to the start of the circular buffer. The coefficient is calculated by the digital LFOs along with the sawtooths required to generate the two pointers.



Algorithm

The digital LFOs are calculated using 24-bit math in the chorus generator block of the DRE. The 20 MSBs (Most Significant Bits) of the results are fed out for use by the address generator and the MAC. Of these 20 bits, the 13 MSBs are routed to the address generator and the 7 LSBs (Least Significant Bits) are used as the interpolation coefficient for the MAC.

The 13 bits routed to the address generator are in two's complement format, giving an effective range of +4095 to -4096. This range is controlled by the amplitude coefficient for the LFO which is 15 bits and always less than 1.

The frequency is controlled by a 13 bit coefficient. These 13 bits are the LSBs of an 18 bit word used internal to the digital LFOs.

The sawtooth waveform frequency is calculated by (assuming a 12.288MHz crystal is being used which results in a 48kHz sample rate):

$$f = (F*F_s*C)/(2*M*H) = 0.045777*F$$

where

F = 13 bit frequency coefficient

F_s = 48000 (sample rate)

C = 4194304 (0x400000, a constant used internally in the LFO)

M = 262143 (0x3FFFF, internal maximum 18-bit value)

H = 8388607 (0x7FFFFFFF, maximum positive 24-bit value)

For a frequency coefficient F of 1, K = 1/262143 and results in f = 0.04577Hz. For a frequency coefficient of 8191 (the maximum 13 bit value), K = 8191/262143 and results in f = 374.9557Hz.

Wavefront Semiconductor ∴ 200 Scenic View Drive ∴ Cumberland, RI 02864 ∴ U.S.A.

Tel: +1 401 658-3670 ∴ Fax: +1 401 658-3680 ∴ Email: info@wavefrontsemi.com

On the web at www.wavefrontsemi.com

Pitch Up

Calculating the proper frequency and amplitude coefficient for pitch shifting is dependent on both length of the circular buffer and the amount of pitch shift desired. We select the buffer to be 8192 samples long (maximum amplitude of the sawtooth waveform) which results in an amplitude coefficient of 32767 (maximum 15 bit amplitude coefficient).

To determine the frequency coefficient, we first need to determine the amount of shift desired. For the case of pitching up 1 octave, we want to play the 8192 samples from the circular buffer at an effective rate of double the sample rate. An easy way to calculate this is to determine how much faster we want to play the samples than the base sample rate. In this case, we want to play the 8192 samples at a rate that is effectively 48kHz added the base sample rate of 48kHz. Since the samples are written to the circular buffer using addresses generated by a down counter, and as the sawtooth is generated by an up counter, we need to invert the sawtooth waveform for the pitch up case. This is done by using the COMPA flag in a CHR instruction. Therefore, to calculate the frequency coefficient, start by finding the sawtooth frequency needed:

$$f = 48\text{kHz faster} / 8192 \text{ samples} = 5.85938\text{Hz}$$

From $f = (F * F_s * C) / (2 * M * \text{SIN})$ above, we can rearrange the equation to solve for F:

$$F = (2 * M * H * f) / (F_s * C) = 127.999, \text{ quantized to } 128.$$

To shift up by amount n, where n is the fraction of an octave, the aggregate formula is:

$$F = [2 * M * H * (2^n - 1)] / (C * \text{BufferSize}) = 1048571.185 * (2^n - 1) / \text{BufferSize}$$

Thus shifting up by 1/2 an octave, using the same 8192 buffer size, gives:

$$F = 1048571.185 * (2^{1/2} - 1) / 8192 = 53.019, \text{ quantized to } 53.$$

Shifting up by 5 semitones gives:

$$F = 1048571.185 * (2^{5/12} - 1) / 8192 = 42.859, \text{ quantized to } 43.$$

Shifting up by 40 cents gives:

$$F = 1048571.185 * (2^{0.40/12} - 1) / 8192 = 2.992, \text{ quantized to } 3.$$

Pitch Down

Pitching down is similar to pitching up, except that rather than adding to the base sample rate, we are subtracting from it. In this case, we do not invert the sawtooth waveform. As an example, if we wished to pitch down by 1 octave, we effectively want to subtract 24kHz from the sample rate. Using the same 8192 sample buffer, we calculate as follows:

$$f = 24\text{kHz slower} / 8192 \text{ samples} = 2.92969\text{Hz}$$

$$F = (2 * M * H * f) / (F_s * C) = 63.9997, \text{ quantized to } 64.$$

To shift down by amount n, where n is the fraction of an octave, the aggregate formula is:

$$F = [2 * M * H * (1 - 2^{-n})] / (C * \text{BufferSize}) = 1048571.185 * (1 - 2^{-n}) / \text{BufferSize}$$

Thus shifting down by 1/2 an octave, using the same 8192 buffer size, gives:

$$F = 1048571.185 * (1 - 2^{-1/2}) / 8192 = 37.490, \text{ quantized to } 37.$$

Shifting down by 5 semitones gives:

$$F = 1048571.185 * (1 - 2^{-5/12}) / 8192 = 32.108, \text{ quantized to } 32.$$

Shifting down by 40 cents gives:

$$F = 1048571.185 * (1 - 2^{-0.40/12}) / 8192 = 2.923, \text{ quantized to } 3.$$

Pitch Down

```

; File: AN3201-02D.ASM
; Description: Example pitch down program
; Authors: Frank Thomson
; Copyright 2001 Wavefront Semiconductor
;
; 1 octave down
LFO0 SAW AMP=32767 FREQ=64 XFAD=1/16
;NOTE: Here we define the waveform:
; SIN, TRI (triangle), SAW (sawtooth for pitch shift)
;
MEM pitchmem 8200 ; Circular buffer memory
MEM temp 1 ; Temp register
MEM temp2 1 ; Temp register
;NOTE: memory locations are referenced by:
; name Start of memory block
; name' End of memory block
; name" Middle of memory block
;
RZP ADCL K=.999 ;0 Read left channel into Accumulator
;
WZP pitchmem K=.999 ;1 Write to the start of the block
; Next read from the middle of the delay memory block using the chorus instruction
; to add the sawtooth waveform to the address. The sample returned is multiplied by
; the 1's comp of the coefficient from the chorus generator block. Use the SIN
; output for the first sawtooth.
CHR0 RZP pitchmem" +SIN COMPK LATCH ;2 Read middle of memory block
;
; Read the sample before the above one and multiply it by the coefficient.
CHR0 RAP pitchmem"+1 +SIN ;3 Read middle -1 from memory block
;
; Save the result to a temp register.
WAP temp K=0 ;4 Write to DAC
;
; Now do it again using the COS output to get the other sawtooth
CHR0 RZP pitchmem" +COS COMPK LATCH ;5
CHR0 RAP pitchmem"+1 +COS ;6
;
; The result from the 2 instructions is in the accumulator, do a write to a temp
; location to bring it back through the multiplier and use the 1's comp of the
; cross fade coefficient to multiply it by.
CHR0 WZP temp2 +COS COMPK MASKA ;7
;
; Get the first result, multiply it by the cross fade coefficient and add it.
CHR0 RAP temp MASKA ;8
WAP OUTL K=0 ;9 Write it to the DAC
;
RZP ADCL K=.999 ;10 Read left channel into Accumulator
WAP OUTR K=0 ;11 Write to DAC
;
;Add a bunch of reads for refresh
RZP 0x00 K=0
RZP 0x40 K=0
RZP 0x80 K=0
RZP 0xc0 K=0
RZP 0x100 K=0
RZP 0x140 K=0
RZP 0x180 K=0
RZP 0x1c0 K=0
RZP 0x200 K=0
RZP 0x240 K=0
RZP 0x280 K=0
RZP 0x2c0 K=0
RZP 0x300 K=0
RZP 0x340 K=0
RZP 0x380 K=0
RZP 0x3c0 K=0

```

NOTICE

Wavefront Semiconductor reserves the right to make changes to their products or to discontinue any product or service without notice. All products are sold subject to terms and conditions of sale supplied at the time of order acknowledgement. Wavefront Semiconductor assumes no responsibility for the use of any circuits described herein, conveys no license under any patent or other right, and makes no representation that the circuits are free of patent infringement. Information contained herein is only for illustration purposes and may vary depending upon a user's specific application. While the information in this publication has been carefully checked, no responsibility is assumed for inaccuracies.

Wavefront Semiconductor products are not designed for use in applications which involve potential risks of death, personal injury, or severe property or environmental damage or life support applications where the failure or malfunction of the product can reasonably be expected to cause failure of the life support system or to significantly affect its safety or effectiveness.

All trademarks and registered trademarks are property of their respective owners.

Contact Information:

Wavefront Semiconductor
200 Scenic View Drive
Cumberland, RI 02864 U.S.A.
Tel: +1 401 658-3670
Fax: +1 401 658-3680
On the web at www.wavefrontsemi.com
Email: info@wavefrontsemi.com

Copyright © 2005 Wavefront Semiconductor

Application note revised March, 2005

Reproduction, in part or in whole, without the prior written consent of Wavefront Semiconductor is prohibited.